

O Elemento *Storage* do *OpenDSS*

Celso Rocha

Mestrando EPUSP - Enerq



VII Simpósio Brasileiro de Sistemas Elétricos

Niterói – RJ

13/05/2018


SBSE 2018

VII Simpósio Brasileiro
de Sistemas Elétricos

Visão Geral

- Modelo Básico
- Perdas por Inatividade
- Estados de Operação
- Modos de Operação
- Informações Adicionais
- Referências

Modelo Básico



Perdas por Inatividade (*Idling Losses*)

- Representam o consumo de energia dos equipamentos auxiliares do armazenador (refrigeração, controle, etc...);
- Modeladas por uma impedância constante;
- Definidas como uma porcentagem da potência nominal (parâmetro *kWRated*) do armazenador;

$$P_n = \frac{|V_n|^2}{R} = \%idlingkW \times kWRated \times 10 \text{ [W]}$$

$$Q_n = \frac{|V_n|^2}{X} = \%idlingkvar \times kWRated \times 10 \text{ [var]}$$

$$R = \frac{|V_n|^2}{\%idlingkW \times kWRated \times 10} \text{ [\Omega]}$$

$$X = \frac{|V_n|^2}{\%idlingkvar \times kWRated \times 10} \text{ [\Omega]}$$

Perdas por Inatividade (*Idling Losses*)

- Representam o consumo de energia dos equipamentos auxiliares do armazenador (refrigeração, controle, etc...);
- Modeladas por uma impedância constante;
- Definidas como uma porcentagem da potência nominal (parâmetro *kWRated*) do armazenador;

$$\bar{Y}_{idling} = \frac{1}{R} + \frac{1}{j \times X} = (\%idlingkW - j \times \%idlingkvar^*) \times \frac{kWRated \times 10}{|V_n|^2} [S]$$

*: na versão mais recente, $\%idlingkvar > 0$ leva à geração de reativos


Perdas por Inatividade (*Idling Losses*)

- Representam o consumo de energia dos equipamentos auxiliares do armazenador (refrigeração, controle, fricção, etc...);
- Modeladas por uma impedância constante;
- Definidas como uma porcentagem da potência nominal (parâmetro *kWRated*) do armazenador;
- Por fim, supondo um elemento armazenador trifásico, a perda por inatividade é trifásica é dada por:

$$\bar{S}_{idling} = \sum_{i=1}^3 \bar{Y}^*_{idling} \times |V_i|^2 \text{ [VA]}$$

Estados de Operação

Estado de Carga



➤ Perdas Carga:


$$Perdas_{ch}[t] = (1 - \eta_{ch}) \times (P_{in}[t] - P_{idling}[t])$$

➤ Perdas Totais:

$$Perdas_{tot}[t] = P_{in}[t] \times (1 - \eta_{ch}) + P_{idling}[t] \times \eta_{ch}$$

Estados de Operação

Estado de Descarga



➤ Perdas Descarga:


$$Perdas_{disch}[t] = (P_{out}[t] + P_{idling}[t]) \times \left(\frac{1}{\eta_{disch}} - 1 \right)$$

➤ Perdas Totais:

$$Perdas_{tot}[t] = P_{out}[t] \times \left(\frac{1}{\eta_{disch}} - 1 \right) + P_{idling}[t] \times \frac{1}{\eta_{disch}}$$

Estados de Operação

- Estado de Inatividade



Supridas pela rede!

Modos de Operação

- 5 Modos: Padrão (*Default*), Nível de Carga (*LoadLevel*), Nível de Preço (*PriceLevel*), Seguidor (*Follow*) e Externo (*External*)
- Regra Geral. O elemento opera em um estado até que:
 - A energia armazenada atinja a capacidade máxima ou a mínima, nos casos de carga e descarga, respectivamente;
 - Algum gatilho seja disparado (válido para os modos que operam com gatilhos);
 - O usuário selecione diretamente um estado através do parâmetro “*state*”;
- Circuito de testes:

Modos de Operação

▪ *Default* (Padrão)

```
! Storage - Modo Default
Clear

New Circuit.Source bus1=A basekv=0.48 phases=3 pu=1

New LoadShape.dispatch_shape interval=1 npts=24
~ mult = [0.386,0.220,0.247,0.280,0.313,0.370,0.589,0.672,0.7477,0.832,0.88,0.94,0.989,0.985,0.98,0.9898,0.999,
1.0,0.958,0.936,0.913,0.800,0.720,0.610]

New Storage.Storage1 phases=3 bus1=A kv=0.48 pf=1 kWrated=50 %reserve=20
~ kWrated= 500 %stored=50 state=idling debugtrace=yes dispmode=default model=1 daily=dispatch_shape
~ ChargeTrigger = 0.34 DischargeTrigger = 0.85

New Monitor.Mon_Storage1_State element=Storage.Storage1 terminal=1 mode=3
New Monitor.Mon_Storage1_Powers element=Storage.Storage1 terminal=1 mode=1 ppolar=No



Set voltagebases = [0.48]
Calc voltagebases

Set mode=Daily
Solve

Plot Monitor object=Mon_Storage1_State channels=(1 2 3 4 5 6 7)
Plot Monitor object=Mon_Storage1_Powers channels=(1 3 5)
```

Modos de Operação

■ Default (Padrão)



$$Perdas_{disch} = (P_{out} + P_{idling}) \times \left(\frac{1}{\eta_{disch}} - 1 \right) = (50 + 0.5) \times \left(\frac{1}{0.9} - 1 \right) = 5.61 kW$$


$$Perdas_{ch} = (P_{in} - P_{idling}) \times \eta_{ch} = (50 - 0.5) \times (1 - 0.9) = 4.95 kW$$

$$Perdas_{tot_{disch}} = 5.61 + 0.5 = 6.11 kW$$

$$Perdas_{tot_{ch}} = 4.95 + 0.5 = 5.45 kW$$

Modos de Operação

▪ Default (Padrão)



147.644kWh

$$Perdas_{tot_{disch}} = 5.61 + 0.5 = 6.11 \text{ kW}$$

$$P_{out_{eff}} = 6.11 + 50 = 56.11 \text{ kW}$$

$$E[17hr] = E[16hr] - 56.11 \text{ kW} \times 1 \text{ hr} = 91.534 \text{ kWh} ???$$

Modos de Operação

■ Follow (Seguidor)

```
! Storage - Modo Seguidor
Clear

New Circuit.Source bus1=A basekv=0.48 phases=3 pu=1

New LoadShape.dispatch_shape interval=1 npts=24
~ mult = [0,-1.0,-1.0,-1.0,-0.5,-0.5,0,0,0,0,0,0,0,0,0.5,0.75,1.0,1.0,1.0,1.0,0.75,0.5,0,0]

New Storage.Storage1 phases=3 bus1=A kv=0.48 pf=1 kWhrated=50 %reserve=20
~ kWhrated= 500 %stored=50 state=idling debugtrace=yes dispmode=follow model=1 daily=dispatch_shape

New Monitor.Mon_Storage1_State element=Storage.Storage1 terminal=1 mode=3
New Monitor.Mon_Storage1_Powers element=Storage.Storage1 terminal=1 mode=1 ppolar=No

Set voltagebases = [0.48]
Calc voltagebases



Set mode=Daily
Solve

Plot Monitor object=Mon_Storage1_State channels=(1 2 3 4 5 6 7)
Plot Monitor object=Mon_Storage1_Powers channels=(1 3 5)
```

Modos de Operação

■ *Follow* (Seguidor)

- $\text{Mult} < 0$: Carga
- $\text{Mult} > 0$: Descarga
- $\text{Mult} = 0$: Inativo
- Potência do *storage* varia com os valores da curva de carga
- Curva só é obedecida se o estado de carga permitir



Modos de Operação

▪ Price e LoadLevel (Preço e Nível de Carga)

```
! Storage - Modo Price
Clear

New Circuit.Source bus1=A basekv=0.48 phases=3 pu=1

New PriceShape.Price interval=1 npts=24
~ price = [75,68,67,69,71,75,75,80,80,80,90,90,90,95,95,95,105,105,110,110,110,90,90,90]

New Storage.Storage1 phases=3 bus1=A kv=0.48 pf=1 kWrated=50 %reserve=20
~ kWhrated= 500 %stored=50 state=idling debugtrace=yes dispmode=price model=1
~ DischargeTrigger = 100 ChargeTrigger= 74

New Monitor.Mon_Storage1.State element=Storage.Storage1 terminal=1 mode=3
New Monitor.Mon_Storage1.Powers element=Storage.Storage1 terminal=1 mode=1 ppolar=No



Set voltagebases = [0.48]
Calc voltagebases

Set pricecurve=Price
Set mode=Daily
Solve

Plot Monitor object=Mon_Storage1.State channels=(1 2 3 4 5 6 7)
Plot Monitor object=Mon_Storage1.Powers channels=(1 3 5)
```


Modos de Operação

▪ *Price e LoadLevel* (Preço e Nível de Carga)



Modos de Operação

- *External*(Controle “Externo”)

```
! Storage – Modo Manual com Reativos especificados com Fator de Potência
Clear

New Circuit.Source bus1=A basekv=0.48 phases=3 pu=1

New Storage.Storage1 phases=3 bus1=A kv=0.48 pf=1 kWrated=50 %reserve=20
~ kWhrated= 500 %stored=50 state=idling debugtrace=yes dispmode=External model=1


New Monitor.Mon.Storage1.State element=Storage.Storage1 terminal=1 mode=3
New Monitor.Mon.Storage1.Powers element=Storage.Storage1 terminal=1 mode=1 ppolar=No

Set voltagebases = [0.48]
Calc voltagebases

Set mode=Daily
Set stepsize=1h
```

Modos de Operação


▪ *External*(Controle “Externo”)



```
// Armazenador ilhado nas duas primeiras horas  
// 01:00hr – 02:00hr  
Set number=2  
Solve
```

Modos de Operação


▪ *External*(Controle “Externo”)



```
// Armazenador carregando nas 5 horas seguintes com uma potência de 80% da potência  
// nominal e um fator de potência de -0.95  
// 03:00hr - 07:00hr  
Edit Storage.Storage1 state=charging %charge=80 pf=-0.95  
Set number=5  
Solve
```

Modos de Operação


▪ *External*(Controle “Externo”)



```
// Armazenador ilhado nas 10 horas seguintes  
// 08:00hr – 17:00hr  
Edit Storage.Storage1 state=idling  
Set number= 10  
Solve
```

Modos de Operação


▪ *External*(Controle “Externo”)



```
// Armazenador descarregando nas 5 horas seguintes com potência nominal e um fator  
// de potência de 0.95  
// 18:00hr – 22:00hr  
Edit Storage.Storage1 state=discharging %discharge=100 pf=0.95  
Set number=5  
Solve
```

Modos de Operação

▪ *External*(Controle “Externo”)




```
// Armazenador ilhado nas 2 horas restantes do dia
// 23:00hr – 24:00hr
Edit Storage.Storage1 state=idling
Set number= 2
Solve
```

Informações Adicionais

■ *StorageController*

- *Controle de múltiplos Armazenadores*
- Modos de Controle:
 - Descarga:
 - PeakShave
 - Follow
 - Support
 - Loadshape (similar ao modo Follow local)
 - Time
 - Schedule
 - Carga:
 - Loadshape
 - Time



Referências

- EPRI: OpenDSS STORAGE Element and STORAGECONTROLLER Element.
<https://sourceforge.net/p/electricdss/code/HEAD/tree/trunk/Distrib/Doc/OpenDSS%20STORAGE%20Element.pdf>. [Online; acessado em 10/05/2018].
- Dugan, R. C., J. A. Taylor e D. Montenegro: Energy Storage Modeling for Distribution Planning. IEEE Transactions on Industry Applications, 53(2):954{962, March 2017, ISSN 0093-9994.
- Rocha, Celso: Analise e Mitigação de Impacos da Conexão de Geração Distribuída, Microgeração Distribuída e Armazenadores em Alimentadores de Distribuição Utilizando o Software OpenDSS.
https://sourceforge.net/p/electricdss/code/HEAD/tree/trunk/Distrib/Examples/Celso_Example/TCC_CelsoRocha.pdf, 2016. [Online; acessado em 10/05/2018].

Obrigado!
Dúvidas?